
1- 概要 

1-1 Visual Basic 6.0 での IVI-COM ドライバの運用 

Visual Basic 6.0 は、IVI-COM 計測器ドライバを使うのに最も適した開発環境の一つです。Visual 
Basic 6.0 では ActiveX コントロールのような COM オブジェクトを使うプログラミング手法が一般に普

及しており、多くの Visual Basic 6.0 プログラマーはこれらを使うことに慣れています。IVI-COM 計測

器ドライバは、ActiveX コントロールではありませんが、一般の COM オブジェクトと同様な手順でプ

ログラムを作成する事が出来ます。 

IVI-COM 計測器ドライバを利用する場合、スペシフィック・インターフェースを利用する方法とクラス・

インターフェースを利用する方法の 2 種類があります。前者は計測器ドライバの固有インターフェー

スを利用するもので、使用する計測器の機能を最大限に利用する事ができます。後者は IVI 仕様

書で定義されている計測器クラスのインターフェースを利用するもので、インターチェンジャビリティ

機能を利用する事ができますが、機種固有の機能を使うことは制限されます。 

Notes: 

計測器ドライバが所属する計測器クラスについては、ドライバ毎の Readme.txt に記載されています。Readme
文書は、Start ボタン Program IVI フォルダから開く事ができます。  

計測器ドライバが如何なる計測器クラスにも属していない場合、クラス・インターフェースを利用する事はできま

せん。つまりこの場合、インターチェンジャビリティ機能を利用するアプリケーションを作成する事は出来ませ

ん。 

1-2 アプリケーション・プロジェクトの作成 

このドキュメントでは、Visual Basic 6.0 で最も一般的なフォーム志向のアプリケーションを例に説明

します。Visual Basic 6.0 統合環境を起動した直後に新規プロジェクトが生成されていない場合は、

File | New Project メニューを選択して New Project ダイアログを表示し、Standard EXE を

選択してアプリケーションのプロジェクトを新規作成してください。 

Notes: 

本ガイドブックでは、IVI-COM Kikusui4800 計測器ドライバ(KIKUSUI PIA4800 シリーズ DC 電源コントローラ)
を使用する例を示します。他機種用の IVI-COM 計測器ドライバでも、ほぼ同様の手順で使用できます。  

2- スペシフィック・インターフェースを使用するサンプル 

ここでは、スペシフィック・インターフェースを使用したサンプルを示します。スペシフィック・インターフ

ェースを使用すると、計測器ドライバで提供される機能を最大限に利用する事ができますが、インタ

ーチェンジャビリティを実現する事はできません。 

2-1 タイプ･ライブラリのインポート 

新規プロジェクトを作成したあと最初にすべき事は、利用したい IVI-COM 計測器ドライバのタイプ・

ライブラリをインポートする事です。Project | References メニューを選択して References ダイ

アログを表示してください。ここでは Kikusui4800 IVI-COM ドライバを使用する例を示すので、

Kikusui4800 (Kikusui) 1.0 Type Library をチェックします。 

 
Figure 2-1 タイプ・ライブラリのインポート 

これでタイプ・ライブラリのインポート作業は完了したので、あなたのアプリケーションからは

Kikusui4800 IVI-COM ドライバで提供されるスペシフィック・インターフェースを使うことができます。 

2-2 オブジェクト・ブラウザ 

タイプ・ライブラリがインポートされると、Visual Basic 6.0 からオブジェクト・ブラウザを通じて利用可

能な COM インターフェースの構文を確認する事ができます。オブジェクト・ブラウザを起動するには、

View | Object Browser メニュー(又は F2 キーを押す)を選択します。 

Figure 2-2 オブジェクト・ブラウザ 

2-3 オブジェクトの作成とセッションのイニシャライズ 

まず、フォーム設計画面の上をマウスでダブルクリックします。すると、Form_Load イベント・ハンド

ラの骨格だけを持つソース・コードが表示されます。フォームのデータ･メンバーとして変数

m_dcpwr を Kikusui4800Lib.Kikusui4800 型で宣言してください。ここでは計測器ドライバ

のオブジェクトを作成するので New 演算子を忘れないように記述します。 

 

Figure 2-3 Form_Load ハンドラ 

オブジェクトを作成しただけでは計測器との I/O は行われていません。計測器との I/O を開始する

には、Initialize メソッドを使用します。Form_Load ハンドラ内に、m_dcpwr. とタイプしてくだ

さい。すると、Visual Basic 6.0 のインテリセンス機能により、Kikusui4800 型に関連するメソッドと

プロパティのリストが現れます。ここでは Initialize メソッドを選択し、Tab キーを押します。 

Figure 2-4 インテリセンス(メソッド・プロパティ・リスト) 

更に、Initialize の後にスペース又は左括弧"("をタイプすると、インテリセンスはこのメソッドの

パラメータを表示します。 

Figure 2-5 インテリセンス(パラメータ・リスト) 

 

ここで、Initialize メソッドのパラメータについて説明しましょう。全ての IVI-COM 計測器ドライ

バは、IVI 仕様書で定義された Initialize メソッドを持っています。このメソッドには、以下のよう

なパラメータがあります。 

Table 2-1 Initialize メソッドのパラメータ 

パラメータ タイプ 説明 

ResourceName String VISA リソース名の文字列。計測器が接続されている I/O イ

ンターフェース、アドレスなどによって決定される。例えば、

GPIB ボード 0 に接続されたプライマリ･アドレス 3 の計測器

であれば、GPIB0::3::INSTR となる。 

IdQuery Boolean TRUE を指定した場合、計測器に対して ID クエリを行う。 

Reset Boolean TRUE を指定した場合、計測器の設定をリセットする。 

OptionString String RangeCheck 
Cache  
Simulate 
QueryInstrStatus 
RecordCoercions 
Interchange Check 

に関する設定を、デフォルト以外に指定できる。更に、計測

器ドライバが DriverSetup 機能をサポートする場合、そ

の設定を行うことができる。 

 

ResourceName には VISA リソースを指定します。IdQuery に TRUE を指定した場合は、計測器

に対して"*IDN?"クエリなどを発行して機種情報を問い合わせます。Reset に TRUE を指定した

場合は、"*RST"コマンドなどを発行して計測器の設定をリセットします。 

OptionString には、2 つの機能があります。1 つは RangeCheck, Cache, Simulate, 

QueryInstrStatus, RecordCoercions, Interchange Check,などの IVI 定義の動

作を設定します。もう 1 つは、計測器ドライバ毎に独自に定義される DriverSetup を指定します。

OptionString は文字列パラメータなので、これらの設定は下のサンプルのような書式でなけれ

ばなりません。 

QueryInstrStatus = TRUE , Cache = TRUE , DriverSetup=12345 

 

設定したい機能の名称及び設定値はケース・インセンシティブ(大文字と小文字の区別なし)です。

設定値は Boolean 型なので、TRUE、FALSE、1、0 の何れかが有効です。複数の項目を設定する

場合は、コンマで区切ります。OptionString パラメータで特に設定値を指定しない場合、IVI 仕
様書で定義されたデフォルト値が適用されます。IVI 仕様書で定義されたデフォルト値は、

RangeCheck と Cache だけが TRUE で、その他は全て FALSE です。 

計測器ドライバによっては、DriverSetup パラメータが意味を持つ場合もあります。これは、IVI
仕様書では定義されない項目を Initialize の呼び出し時に指定するもので、利用目的や書式

はドライバ依存です。従って DriverSetup の指定を行う場合、それは OptionString の最後

の項目として指定される必要があります。DriverSetup の指定内容はドライバ毎に異なるので、

ドライバの Readme 文書又はオンライン・ヘルプなどを参照してください。 

では具体的に、Initialize メソッドの呼び出しを記述してみましょう。OptionString パラメー

タはオプションなので、ここでは省略しています。 

Private Sub Form_Load() 
    m_dcpwr.Initialize "GPIB0::3::INSTR", True, True 
End Sub 

2-4 セッションのクローズ 

計測器ドライバによるセッションをクローズするには、Close メソッドを使います。ここでの例では、

Initialize メソッドの呼び出しを Form_Load ハンドラに記述したので、Close メソッドの呼び

出しは Form_Unload に書くのが良いでしょう。 

 

Option Explicit 
Private m_dcpwr As New Kikusui4800Lib.Kikusui4800 
 
Private Sub Form_Load() 
    m_dcpwr.Initialize "GPIB0::3::INSTR", True, True 
End Sub 
 
Private Sub Form_Unload(Cancel As Integer) 
    m_dcpwr.Close 
End Sub 

 

2-5 実行 

ここまでのコードだけで、とりあえず実行する事は可能です。この例では Form_Load ハンドラ内で

Initialize メソッドの呼び出しが行われるので、プログラムを実行すると、即座に計測器との通

信が開始されます。実際に計測器が接続されていて Initialize メソッドが成功した場合は、フォ

ーム画面が表示されます。通信に失敗した場合や、VISA ライブラリの設定が正しく行われていない

場合などは、COM 例外(Visual Basic 6.0 ランタイム・エラー)を発生します。 

 

Figure 2-6 COM 例外 

2-6 リピーテッド・キャパビリティ 

Kikusui4800 IVI-COM ドライバの場合、IviDCPwr クラスで定義されているコンセプトと同様に、DC
電源装置の出力設定は Output インターフェースを通じて行われます。Kikusui4800 ドライバで提供

されるスペシフィック・インターフェースの場合、IKikusui4800Output と

IKikusui4800Outputs がそうです。IviDCPwr クラスに属する計測器ドライバは、複数の出力チ

ャンネルを持つマルチ・トラック電源装置を前提に設計されています。 

これらの COM インターフェースは、単数形と複数形の違いを除いて、同じ名前になっています。この

ように複数形の名前を持つインターフェースは、IVI 仕様書では一般にリピーテッド・キャパビリティと

呼ばれます。リピーテッド・キャパビリティとは、機能が全く同じ又は類似している複数のオブジェクト

を扱うために定義されたコンテナのようなもので、IKikusui4800Outputs のような複数形の名

前を持つ COM インターフェースは通常 Count、Name、Item プロパティ(いずれもリード・オンリー)
を持ちます。また、Item プロパティを通じて単品のオブジェクトを参照する事もできます。 

まずは、下記の例を見てください。これは、Kikusui4800 IVI-COM ドライバがサポートする電源装置

(実際には Kikusui PIA4800 シリーズ電源コントローラ)の"N5!C1"で識別される出力チャンネルを

制御するものです。この例では、フォームにボタン(Command1)を貼り付けた場合の、イベント・ハン

ドラとして記述しています。 

Private Sub Command1_Click() 
    Dim out As Kikusui4800Lib.IKikusui4800Output 
    Set out = m_dcpwr.Outputs.Item("N5!C1") 
     
    out.VoltageLevel = 10.5 
    out.CurrentLimit = 1.2 
    out.Enabled = True  
End Sub 

一旦 IKikusui4800Output インターフェースを取得してしまえば、あとは難しい事はありません。

VoltageLevel プロパティは電圧レベル設定を、CurrentLimit プロパティは電流リミット設定

を、それぞれ行います。Enabled プロパティは出力の ON/OFF 設定を行います。 

IKikusui4800Output インターフェースを取得する際の記述に注意してください。ここでは、

IKikusui4800 インターフェースの Outputs プロパティを通じて IKikusui4800Outputs を

取得し、直ぐに Item プロパティを使って IKikusui4800Output インターフェースを取得してい

ます。 

    Dim out As Kikusui4800Lib.IKikusui4800Output 
    Set out = m_dcpwr.Outputs.Item("N5!C1") 

 

このコードは次のように書くこともできます。 

    Dim outs As Kikusui4800Lib.IKikusui4800Outputs 
    Dim out As Kikusui4800Lib.IKikusui4800Output 
    Set outs = m_dcpwr.Outputs 
    Set out = outs.Item("N5!C1") 

 

ここで、Item プロパティに渡しているパラメータに注意する必要があります。このパラメータは参照

したい単品の Output オブジェクトの名前を指定しています。しかしここで使える名前(Output 
Name)は、ドライバごとにそれぞれ違います。例えば Kikusui4800 IVI-COM ドライバでは、

"N1!C1"のような、NODE と CH を指定する表現になっていますが、他のドライバでは(たとえ、同じ

IviDCPwr クラスであっても)違ったものになります。例えば、他の計測器ドライバでは、"Track1"

のような表現かも知れません。特定の計測器ドライバで使用可能は名前は、通常はドライバのオン

ライン・ヘルプなどに記載されていますが、下記のようなテスト・コードを書くことでそれらを調べる事

も可能です。 

    Dim outs As Kikusui4800Lib.IKikusui4800Outputs 
    Set outs = m_dcpwr.Outputs 
     
    Dim cnt As Long 
    Dim ndx As Long 
    cnt = outs.Count 
    For ndx = 1 To cnt 
        Dim strName As String 
        strName = outs.Name(ndx) 
        Debug.Print strName 
    Next ndx 

 

Count プロパティは、リピーテッド・キャパビリティが持つ単品オブジェクトの個数を返します。Name

プロパティは、与えられたインデックス番号の単品オブジェクトが持つ名前を返します。この名前こそ

が、Item プロパティに渡す事のできるパラメータになるのです。上記の例では、For/Next ステー

トメントを使って、インデックス 1 から Count までを反復処理しています。Name パラメータに渡すイ

ンデックス番号は 0 ベースではなく 1 ベースである事に注意してください。 

3- クラス・インターフェースを使用するサンプル 

ここでは、クラス・インターフェースを使用したサンプルを示します。クラス・インターフェースを使用す

ると、アプリケーションを再度コンパイル・リンクすることなく、計測器を交換する事ができます。但し

その場合、交換前後の両機種に対して IVI-COM 計測器ドライバが提供されており、且つそれらのド

ライバが同じ計測器クラスに属している必要があります。異なる計測器クラス間でのインターチェン

ジャビリティは実現できません。 

3-1 仮想インストルメント 

インターチェンジャビリティ機能を利用するアプリケーションの作成を行う前にやっておかなければな

らない事は、仮想インストルメントの作成です。インターチェンジャビリティ機能を実現するには、アプ

リケーション・コード内に特定の IVI-COM 計測器ドライバに依存した記述(例えば Kikusui4800 型で

直接オブジェクトを生成)したり、"GPIB0::3::INSTR"のような特定 VISA リソース名の記述などをす

るべきではありません。これらの事柄をアプリケーション内に直接記述すると、インターチェンジャビ

リティを損ないます。 

その代わりに、IVI-COM 仕様では、計測器ドライバとアプリケーションの外部に IVI コンフィグレーシ

ョン・ストアを置く事によってインターチェンジャビリティを実現します。アプリケーションは IVI コンフィ

グレーション・ストアの内容に従って計測器ドライバの選択を間接的に行い、間接的にロードされた

計測器ドライバを特定機種に依存しないクラス・インターフェースを通じてアクセスします。 

IVI コンフィグレーション・ストアは通常、/Program Files/IVI/Data/IviConfigurationStore.XML ファ

イルで、IVI Configuration Server DLL を通じてアクセスされます。この DLL を利用するのは、主に

IVI-COM 計測器ドライバと計測器ドライバのベンダーによって提供されるコンフィグレーション・ツー

ルであって、アプリケーションからは通常は使いません。弊社の場合はコンフィグレーション･ツール

として、Kikusui IVI Config Utility を提供しています。これを利用する事で、仮想インストルメント

の設定を行うことが出来ます。 

Notes: 

Kikusui IVI Config Utility を使用して仮想インストルメントの設定を行う手順に関しては、「プログラミング・ガイ

ド(IVI Config Utility 編)」を参照してください。 

このガイドブックでは、MySupply というロジカル・ネームで仮想インストルメントが既に作成されて

いて、Kikusui4800 ドライバを使用し、VISA リソース"GPIB0::3::INSTR"を使用する、という設定が

行われているものとします。 

3-2 タイプ･ライブラリのインポート 

新規プロジェクトを作成したあと最初にすべき事は、利用したい IVI-COM インターフェース･クラスの

タイプ・ライブラリをインポートする事です。Project | References メニューを選択して

References ダイアログを表示してください。ここでは IviDCPwr クラス・インターフェースを使用する

ので、IviDCPwr 2.0 TypeLibrary をチェックします。また、IviDriver 1.0 Type Library と

IviSessionFactory 1.0 TypeLibrary の 2 つは、使用する計測器クラス・インターフェースに関

係なく、必ずチェックしてください。 

 
Figure 3-1 タイプ・ライブラリのインポート 

これでタイプ・ライブラリのインポート作業は完了したので、あなたのアプリケーションからは任意の

計測器ドライバを IviDCPwr クラス・インターフェースを通じて使うことができます。 

3-3 オブジェクト・ブラウザ 

タイプ・ライブラリがインポートされると、Visual Basic 6.0 からオブジェクト・ブラウザを通じて、利用

可能な COM インターフェースの構文を確認する事ができます。オブジェクト・ブラウザを起動するに

は、View | Object Browser メニュー(又は F2 キーを押す)を選択します。 

Figure 3-2 オブジェクト・ブラウザ 

3-4 オブジェクトの作成とセッションのイニシャライズ 

まず、フォーム設計画面の上をマウスでダブルクリックします。すると、Form_Load イベント・ハンド

ラの骨格だけを持つソース・コードが表示されます。フォームのデータ･メンバーとして変数

m_dcpwr を IviDCPwrLib.IIviDCPwr 型で宣言してください。IIviDCPwr のように大文字の

"I"で始まるタイプは一般に COM インターフェース型なので、New 演算子でオブジェクトを作成する

事はできません。 

 

Figure 3-3 Form_Load ハンドラ 

次に、計測器ドライバのオブジェクトを作成しなければなりません。計測器ドライバ・オブジェクトの作

成には、IVI Session Factory を使用します。IVI Session Factory は IVI Shared Components によ

って提供される COM DLL サーバーで、指定されたロジカル・ネームから該当する仮想インストルメ

ントのコンフィグレーション情報を抽出し、適切な計測器ドライバ・ソフトウェアをロードしてオブジェク

トを作成するものです。 

Private Sub Form_Load() 
     
    Dim sf As New IVISESSIONFACTORYLib.IviSessionFactory 

    Set m_dcpwr = sf.CreateDriver("MySupply") 
 
    m_dcpwr.Initialize "MySupply", True, True 
     
End Sub 

 

ここでは、IviSessionFactory オブジェクトを作成し、CreateDriver メソッドで計測器オブジ

ェクトを作成しています。パラメータに指定されている"MySupply"は、有効なロジカル・ネームとし

てコンフィグレーションが行われている必要があります。また、この時にロードされる計測器ドライバ

DLL は、仮想インストルメント MySupply で指定されたものが自動的に割り当てられます。 

ドライバ・オブジェクトが作成できたら、Initialize メソッドを呼び出します。Initialize メソッ

ドのパラメータは、スペシフィック・インターフェースを使用した場合と全く同じですが、

ResourceName パラメータには、VISA リソースを直接指定するのではなく、ロジカル・ネームを指

定します。 

ここで、Initialize メソッドのパラメータについて再度説明しましょう。全ての IVI-COM 計測器ド

ライバは、IVI 仕様書で定義された Initialize メソッドを持っています。このメソッドには、以下

のようなパラメータがあります。 

Table 3-1 Initialize メソッドのパラメータ 

パラメータ タイプ 説明 

ResourceName String VISA リソース名の文字列。計測器が接続されている I/O イ

ンターフェース、アドレスなどによって決定される。例えば、

GPIB ボード 0 に接続されたプライマリ･アドレス 3 の計測器

であれば、GPIB0::3::INSTR となる。 

ロジカル・ネームを指定した場合、そのロジカル･ネームの

Hardware Asset コンフィグレーションに従った VISA リソー

スが間接指定される。 

IdQuery Boolean TRUE を指定した場合、計測器に対して ID クエリを行う。 

Reset Boolean TRUE を指定した場合、計測器の設定をリセットする。 

OptionString String RangeCheck 
Cache  
Simulate 
QueryInstrStatus 
RecordCoercions 
Interchange Check 

に関する設定を、デフォルト以外に指定できる。更に、計測

器ドライバが DriverSetup 機能をサポートする場合、そ

の設定を行うことができる。 

 

インターチェンジャブル・アプリケーションでは通常、ResourceName には VISA リソースではなくロ

ジカル･ネームを指定します。実際には VISA リソースを直接指定する事もできますが、仮想インスト

ルメントの抽象性が少し損なわれます。 

IdQuery に TRUE を指定した場合は、計測器に対して"*IDN?"クエリなどを発行して機種情報を

問い合わせます。Reset に TRUE を指定した場合は、"*RST"コマンドなどを発行して計測器の設

定をリセットします。 

OptionString には、2 つの機能があります。1 つは RangeCheck, Cache, Simulate, 

QueryInstrStatus, RecordCoercions, Interchange Check,などの IVI 定義の動

作を設定します。もう 1 つは、計測器ドライバ毎に独自に定義される DriverSetup を指定します。

OptionString は文字列パラメータなので、これらの設定は下のサンプルのような書式でなけれ

ばなりません。 

QueryInstrStatus = TRUE , Cache = TRUE , DriverSetup=12345 

 

設定したい機能の名称及び設定値はケース・インセンシティブ(大文字と小文字の区別なし)です。

設定値は Boolean 型なので、TRUE、FALSE、1、0 の何れかが有効です。複数の項目を設定する

場合は、コンマで区切ります。OptionString パラメータで特に設定値を指定しない場合に適用さ

れるデフォルト値は、ResourceName にロジカル・ネームを指定した場合と VISA リソースを直接指

定した場合とで異なります。ResourceName にロジカル・ネームを指定した場合のデフォルトは、そ

のロジカル・ネームで示される仮想インストルメントの Default Operation で指定されたものになりま

す。一方 VISA リソースを直接指定した場合は、ロジカル・ネームをバイパスしてしまうので、IVI 仕
様書で定義されたデフォルト値が適用されます。IVI 仕様書で定義されたデフォルト値は、

RangeCheck と Cache だけが TRUE で、その他は全て FALSE です。 

計測器ドライバによっては、DriverSetup パラメータが意味を持つ場合もあります。これは、IVI
仕様書では定義されない項目を Initialize の呼び出し時に指定するもので、利用目的や書式

はドライバ依存です。従って DriverSetup の指定を行う場合、それは OptionString の最後

の項目として指定される必要があります。DriverSetup の指定内容はドライバ毎に異なるので、

ドライバの Readme 文書又はオンライン・ヘルプなどを参照してください。 

では具体的に、Initialize メソッドの呼び出しを記述してみましょう。OptionString パラメー

タはオプションなので、ここでは省略しています。 

 
    m_dcpwr.Initialize "MySupply", True, True 
 

3-5 セッションのクローズ 

計測器ドライバによるセッションをクローズするには、Close メソッドを使います。ここでの例では、

Initialize メソッドの呼び出しを Form_Load ハンドラに記述したので、Close メソッドの呼び

出しは Form_Unload に書くのが良いでしょう。 

 

Option Explicit 
Private m_dcpwr As IviDCPwrLib.IIviDCPwr 
 
Private Sub Form_Load() 
     
    Dim sf As New IVISESSIONFACTORYLib.IviSessionFactory 
    Set m_dcpwr = sf.CreateDriver("MySupply") 
 
    m_dcpwr.Initialize "MySupply", True, True 
     
End Sub 
 
Private Sub Form_Unload(Cancel As Integer) 
    m_dcpwr.Close 
End Sub 

 

3-6 実行 

ここまでのコードだけで、とりあえず実行する事は可能です。この例では Form_Load ハンドラ内で

Initialize メソッドの呼び出しが行われるので、プログラムを実行すると、即座に計測器との通

信が開始されます。実際に計測器が接続されていて Initialize メソッドが成功した場合は、フォ

ーム画面が表示されます。通信に失敗した場合や、VISA ライブラリの設定が正しく行われていない

場合などは、COM 例外(Visual Basic 6.0 ランタイム・エラー)を発生します。 

Figure 3-4 COM 例外 

3-7 リピーテッド・キャパビリティ 

IviDCPwr クラス・インターフェースでは、DC 電源装置の出力設定は Output インターフェースを通じ

て行われます。ここで使用するインターフェースは、IIviDCPwrOutput と IIviDCPwrOutputs

です。IviDCPwr クラスに属する計測器ドライバは、複数の出力チャンネルを持つマルチ・トラック電

源装置を前提に設計されています。 

これらの COM インターフェースは、単数形と複数形の違いを除いて、同じ名前になっています。この

ように複数形の名前を持つインターフェースは、IVI 仕様書では一般にリピーテッド・キャパビリティと

呼ばれます。リピーテッド・キャパビリティとは、機能が全く同じ又は類似している複数のオブジェクト

を扱うために定義されたコンテナのようなもので、IIviDCPwrOutputs のような複数形の名前を

持つ COM インターフェースは通常 Count、Name、Item プロパティ(いずれもリード・オンリー)を持

ちます。また、Item プロパティを通じて単品のオブジェクトを参照する事もできます。 

まずは、下記の例を見てください。これは、仮想インストルメントのバーチャル･ネームに登録されて

いる"Track_A"で識別される出力チャンネルを制御するものです。この例では、フォームにボタン

(Command1)を貼り付けた場合の、イベント・ハンドラとして記述しています。 

Private Sub Command1_Click() 
    Dim out As IviDCPwrLib.IIviDCPwrOutput 
    Set out = m_dcpwr.Outputs.Item("Track_A") 
     
    out.VoltageLevel = 10.5 
    out.CurrentLimit = 1.2 
    out.Enabled = True 
End Sub 

一旦 IIviDCPwrOutput インターフェースを取得してしまえば、あとは難しい事はありません。

VoltageLevel プロパティは電圧レベル設定を、CurrentLimit プロパティは電流リミット設定

を、それぞれ行います。Enabled プロパティは出力の ON/OFF 設定を行います。 

IIviDCPwrOutput インターフェースを取得する際の記述に注意してください。ここでは、

IIviDCPwr インターフェースの Outputs プロパティを通じて IIviDCPwrOutputs を取得し、

直ぐに Item プロパティを使って IIviDCPwrOutput インターフェースを取得しています。 

    Dim out As IviDCPwrLib.IIviDCPwrOutput 
    Set out = m_dcpwr.Outputs.Item("Track_A") 

 

このコードは次のように書くこともできます。 

    Dim outs As IviDCPwrLib.IIviDCPwrOutputs 
    Dim out As IviDCPwrLib.IIviDCPwrOutput 
    Set outs = m_dcpwr.Outputs 
    Set out = outs.Item("Track_A") 

 

ここで、Item プロパティに渡しているパラメータに注意する必要があります。このパラメータは参照

したい単品の Output オブジェクトの名前を指定しています。スペシフィック・インターフェースを使用

した例ではドライバごとにそれぞれ異なる名前(フィジカル・ネーム)を直接していしましたが、ここで

は違います。ここでは特定の計測器ドライバに依存したフィジカル・ネームは使えないので、バーチ

ャル・ネームを指定します。 

3-8 計測器の交換 

これまでの例では、仮想インストルメントのコンフィグレーションとして、Kikusui4800 計測器ドライバ

を設定しましたが、ここで計測器を例えば AgilentE36xx ドライバでホストされるものに交換するとど

うなるでしょう。その場合には、アプリケーションを再度コンパイル・リンクする必要はありませんが、

使用する仮想インストルメントのコンフィグレーションを変更する必要があります。 

変更しなければならないコンフィグレーションは、基本的には、Driver Session タブにある Software 
Module の選択と、Virtual Names タブでのバーチャル・ネームのマップ変更(マップ先のフィジカル･

ネームが変わるため)です。計測器が変わると、必ずしも同じ I/O インターフェースを使えるとは限ら

ないので(GPIB オンリーの計測器から RS232 オンリーの計測器へのチェンジなど)、必要であれば、

Hardware Asset タブにある IO Resource Descriptor も設定変更してください。 

Notes: 

仮想インストルメントのコンフィグレーション方法については、「計測器ドライバ・プログラミング･ガイド(IVI 
Config Utility 編)」を参照してください。 

4- エラー処理 

これまで示したサンプルでは、エラー処理を何も行っていませんでした。しかし実際には、範囲外の

値をプロパティに設定したり、サポートされていない機能を呼び出したりすると、計測器ドライバがエ

ラーを発生する事があります。また、どんなに堅牢に設計・実装されたアプリケーションでも、計測器

との I/O 通信エラーは避けることが出来ません。 

IVI-COM 計測器ドライバでは、計測器ドライバ内で発生したエラーは全て COM 例外としてクライア

ント・プログラムに伝えられます。Visual Basic 6.0 の場合、COM 例外は On Error Goto ステー

トメントを使って処理する事が出来ます。 

先ほど示した、電圧・電流を設定するコードを下記のように変更してみましょう。 

Private Sub Command1_Click() 
    On Error GoTo DRIVER_ERR: 
 
    Dim out As Kikusui4800Lib.IKikusui4800Output 
    Set out = m_dcpwr.Outputs.Item("N5!C1") 
     
    out.VoltageLevel = 10.5 
    out.CurrentLimit = 1.2 
    out.Enabled = True 
     
    Exit Sub 
DRIVER_ERR: 
    MsgBox Err.Description, vbOKOnly, "Err 0x" & Hex(Err.Number) 
End Sub 

 

ここでは On Error Goto ステートメントを使ってエラー処理を行っています。例えば、Item プロ

パティに渡した名前が間違っている場合、VoltageLevel に設定する値が適正範囲から外れてい

る場合、或いは計測器との通信に失敗した場合などはいずれも、計測器ドライバ内で COM 例外が

発生します。上記の例では、例外が発生した場合に簡単なメッセージ・ボックスを表示しています。 

エラー(COM 例外)の詳細は、Visual Basic 6.0 で定義されている Err オブジェクトを通じて取得で

きます。この例では、メッセージ･ボックスのキャプションを Number プロパティから取得したエラー・

コード(16 進表記)に、Description プロパティから取得した文字列を本文に、それぞれ設定して

います。 

 

Figure 4-1 エラー処理によるメッセージ・ボックス 

